Pensieve header: Working in the double of the 2D pencil, as determined in “Doubling.nb”; continues pensieve://2017-06/.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\2017-08"];
```

The “degree carrier” is $\hbar$, and all “coupling constants” are proportional to it.

```
(*$TD=∞; ℏ /: ℏ^d_· /;d>$TD :=0;*)
$TϵD = 2; ϵ /: ϵ^d_· /; d > $TϵD := 0;
```

The Doubled Algebra $\mathcal{U}_{\hbar\gamma\epsilon}$

Change relative to “Doubling.nb”: We use $t=\epsilon a-\gamma b$ and $T=e^{\hbar t}=A^{-1}B$ instead of b and B.

Implementing $\mathcal{U}_{\hbar\gamma\epsilon}$

With $q=e^{\hbar\gamma\epsilon}$, $A=e^{-\hbar\epsilon a}$, and $[f,g]_q:=fg-qgf$, our algebra is $\mathcal{U}_{\hbar\gamma\epsilon}=\langle t,y,a,x\rangle/\mathcal{R}$, where $\mathcal{R}=\left([t,*]=0,\ [a,y]=-\gamma y,\ [x,y]_q=\hbar^{-1}(1-TA^2),\ [x,a]=-\gamma x\right)$.

```
q := Sum[(ℏ γ ϵ)^k / k!, {k, 0, Min[$TϵD]}];
AlgebraAtom = y | a | x;
PBWRule = {y → 1, a → 2, x → 3};
B[U@a, U@y] = -γ U@y; B[U@x, U@a] = -γ U@x;
(*B[U@x,U@y]=(q-1) UU[y,x]+UU@Sum[-(t-2 a ϵ)^(k+1) ℏ^k/(k+1)! ,{k,0,$TD}];*)
B[U@x, U@y] := (q - 1) UU[y, x] + UU[ℏ^-1 (1 - Sum[T (-2 a ϵ ℏ)^k / k! , {k, 0, $TϵD}])];
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule]; x_ ≺ y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
U_i_[ℰ_] := ℰ /. {t → t_i, T → T_i, u_U :> Replace[u, x_ :> x_i, 1]};
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, x_] = 0;
B[U[y_], U[x_]] = Simp[-B[U[x], U[y]]];
B[x_, y_] := x ** y - y ** x;
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

```
U[xx___, x_] ** U[y_, yy___] :=
  If[x ≤ y, U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
```

```
UU[c_. * (l : AlgebraAtom)^n_, r___] /; FreeQ[c, AlgebraAtom] :=
  Expand[c UU[Sequence @@ Table[l, {n}], r]];
UU[c_. * l : AlgebraAtom, r___] := Expand[c U[l] ** UU[r]];
UU[c_, r___] /; FreeQ[c, AlgebraAtom] := Expand[c UU[r]];
UU[] = U[];
UU[l_Plus, r___] := UU[#, r] & /@ l;
UU[l_, r___] := UU[Expand[l], r];
```

```
O[poly_, specs___] := Module[{vs, us, z},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) :> (l /. x_i_ :> x_s));
  Simp@Total[CoefficientRules[Normal@Series[poly, {ħ, 0, $TD}], vs] /.
     (p_ → c_) :> c UU @@ (us^p)]
 ]
```

```
$TϵD = 5; B[U@x, U@y] // Simp
```

$$\left(\frac{1}{\hbar}-\frac{\mathrm{T}}{\hbar}\right)\mathrm{U}[\,]+2\,\mathrm{T}\,\epsilon\,\mathrm{U}[\mathrm{a}]-2\,\mathrm{T}\,\epsilon^2\,\hbar\,\mathrm{U}[\mathrm{a},\mathrm{a}]+$$
$$\left(\gamma\,\epsilon\,\hbar+\frac{1}{2}\gamma^2\epsilon^2\hbar^2+\frac{1}{6}\gamma^3\epsilon^3\hbar^3+\frac{1}{24}\gamma^4\epsilon^4\hbar^4+\frac{1}{120}\gamma^5\epsilon^5\hbar^5\right)\mathrm{U}[\mathrm{y},\mathrm{x}]+$$
$$\frac{4}{3}\mathrm{T}\,\epsilon^3\,\hbar^2\,\mathrm{U}[\mathrm{a},\mathrm{a},\mathrm{a}]-\frac{2}{3}\mathrm{T}\,\epsilon^4\,\hbar^3\,\mathrm{U}[\mathrm{a},\mathrm{a},\mathrm{a},\mathrm{a}]+\frac{4}{15}\mathrm{T}\,\epsilon^5\,\hbar^4\,\mathrm{U}[\mathrm{a},\mathrm{a},\mathrm{a},\mathrm{a},\mathrm{a}]$$

```
$TϵD = 2; z1 = U[y, y, a, a, x, x]; z2 = U[y, a, x]; z3 = U[y, y, a, x];
z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp
```

0

The Logos and Testing

From Classes:17-1350-AKT:170317-g1Invariant@.nb:

```
Λ_k_[h_, e_, l_, f_, α_, β_, δ_] := Module[
   {ρh, ρe, ρl, ρf, eqn, a, b, c, sol, λ, q, ν},
   ρh = ( -1  0 ; 0  -1 );  ρe = ( 0  0 ; -ϵ  0 );  ρl = ( -(1 + 1/ϵ)/2   0 ; 0   (1 - 1/ϵ)/2 );  ρf = ( 0  1 ; 0  0 );
   eqn = MatrixExp[α ρf].MatrixExp[β ρe] ==
     MatrixExp[a ρe].MatrixExp[c (ρh - 2 ϵ ρl)].MatrixExp[b ρf];
   Echo[sol = Solve[Thread[Flatten /@ eqn], {a, b, c}][[1]] /. C[1] → 0];
   λ = Simplify[e^(-f α - e β + h α β) Normal@Series[e^(c h + a e - 2 ϵ c l + b f) /. sol, {ϵ, 0, k}]];
   q = e^(ν (f α + e β - h α β + e f δ));
   Collect[q^-1 DP_{α→D_f, β→D_e}[λ][q] /. ν → (1 + h δ)^-1, ϵ, Simplify]];
```